This Maple worksheet is denoted sirs2c.
It is used to study an SIRS model with demography, for $R0>1$ and $\alpha 1$ large.
The deterministic version of the model then shows damped oscillations toward an endemic infection level.
The particular model dealt with here has a state space with 2 variables: S and I.
This means that the variable R is approximated by N-S-I.
Furthermore, the infection rate is "conventional": the denominator S+I+R-1 of the proper infection rate is replaced by N.
Two things are done here:
I) We derive an expression for the angular frequency of the deterministic model oscillations; and
II) We give a derivation of the moments of the stationary distribution of a diffusion approximation.
Ingemar Nåsell, KTH, Stockholm, 2012-10-03.

```
> restart;
  with(LinearAlgebra,Transpose,Eigenvalues,
  CharacteristicPolynomial);
  with(VectorCalculus,Jacobian);
  interface(imaginaryunit=II);
```

$$[Transpose, Eigenvalues, CharacteristicPolynomial]$$

$$[Jacobian]$$

$$I \quad (1)$$

The reason for changing the notation used for the imaginary unit is that "I" will be used below to denote the number of infected individuals.
The original transition rates are stored in the table transA:

```
> transA:=table([[1,0]=mu*N+delta*(N-S-I),[-1,1]=beta*S*I/N,[-1,
  0]=mu*S,[0,-1]=(mu+gamma)*I]):
```

After scaling: x1=$\frac{S}{N}$, x2=$\frac{I}{N}$, and reparametrization R0=$\frac{\beta}{\gamma+\mu}$, $\alpha 1=\frac{\gamma+\mu}{\mu}$, $\alpha 2=\frac{\delta+\mu}{\mu}$,
we get: S=$x1\cdot N$, I=$x2\cdot N$, $\beta=\mu\cdot\alpha 1\cdot R0$, $\gamma=\mu\cdot(\alpha 1-1)$, $\delta=\mu\cdot(\alpha 2-1)$.
The Maple procedure "scale" is used to change the transition rates accordingly:

```
> scale:=proc(tab)
    local xA,n,xB,xC;
    xA:=op(2,eval(tab));
    n:=nops(xA);
    xB:=subs(S=x1*N,I=x2*N,beta=mu*alpha1*R0,gamma=mu*(alpha1-1),
  delta=mu*(alpha2-1),xA);
    xC:=[seq(lhs(op(i,xB))=simplify(rhs(op(i,xB)/N)),i=1..n)];
    table(xC);
  end proc:
```

We apply the scaling and reparametrization mentioned above to get the table of transition rates "trans":

```
> trans:=scale(transA);
```

$$trans := table([[0,-1]=\mu\,\alpha 1\,x2,\ [-1,0]=\mu\,x1,\ [-1,1]=\mu\,\alpha 1\,R0\,x1\,x2,\ [1,0]=-\mu\,(-\alpha 2+\alpha 2\,x1+\alpha 2\,x2-x1-x2)]) \quad (2)$$

Next is a procedure that determines the right-hand sides of the deterministic ODEs for the scaled variables x1 and x2 from the table of transition rates:

```
> equ:=proc(i,tab)
    local x,n;
    x:=op(2,eval(tab));
    add(lhs(x[n])[i]*rhs(x[n]),n=1..nops(x));
  end proc:
```

The right-hand sides of the ODEs for the variables x1 and x2 are found to be:

```
> eq1:=simplify(equ(1,trans));
  eq2:=simplify(equ(2,trans));
```

$$eq1 := -\mu\,\alpha 1\,R0\,x1\,x2 + \mu\,\alpha 2 - \mu\,\alpha 2\,x1 - \mu\,\alpha 2\,x2 + \mu\,x2$$

$$eq2 := -\mu\,\alpha 1\,x2 + \mu\,\alpha 1\,R0\,x1\,x2 \tag{3}$$

Now determine the critical points of these two ODEs:

```
> crit:=solve({eq1,eq2},{x1,x2});
```

$$crit := \{x1 = 1, x2 = 0\}, \left\{x1 = \frac{1}{R0}, x2 = \frac{\alpha 2\,(R0-1)}{R0\,(\alpha 1+\alpha 2-1)}\right\} \tag{4}$$

The point corresponding to an endemic infection level is termed (x10,x20):

```
> x10:=rhs(crit[2][1]);
  x20:=rhs(crit[2][2]);
```

$$x10 := \frac{1}{R0}$$

$$x20 := \frac{\alpha 2\,(R0-1)}{R0\,(\alpha 1+\alpha 2-1)} \tag{5}$$

The Jacobian of the system of ODEs is denoted Bx:

```
> Bx:=Jacobian([eq1,eq2],[x1,x2]);
```

$$Bx := \begin{bmatrix} -\mu\,\alpha 1\,x2\,R0 - \mu\,\alpha 2 & -\mu\,\alpha 1\,R0\,x1 - \mu\,\alpha 2 + \mu \\ \mu\,\alpha 1\,x2\,R0 & -\mu\,\alpha 1 + \mu\,\alpha 1\,R0\,x1 \end{bmatrix} \tag{6}$$

Evaluate the Jacobian at the critical point:

```
> B:=simplify(subs(x1=x10,x2=x20,Bx));
```

$$B := \begin{bmatrix} -\frac{\mu\,\alpha 2\,(\alpha 1\,R0+\alpha 2-1)}{\alpha 1+\alpha 2-1} & -\mu\,\alpha 1 - \mu\,\alpha 2 + \mu \\ \frac{\mu\,\alpha 1\,\alpha 2\,(R0-1)}{\alpha 1+\alpha 2-1} & 0 \end{bmatrix} \tag{7}$$

We proceed to determine the eigenvalues of the matrix B.
The command "CharacteristicPolynomial" gives

```
> p:=CharacteristicPolynomial(B,lambda);
```

$$p := \lambda^2 + \frac{\mu\,\alpha 2\,(\alpha 1\,R0+\alpha 2-1)\,\lambda}{\alpha 1+\alpha 2-1} + (R0-1)\,\alpha 2\,\alpha 1\,\mu^2 \tag{8}$$

The eigenvalues are found as solutions of the characteristic equation p=0 or $\lambda^2 + a\cdot\lambda$

+ b = 0.

They can be written $-\frac{a}{2} \pm i\cdot\Omega$, with $\Omega = \sqrt{b - \left(\frac{a}{2}\right)^2}$ =

$\mu\cdot\sqrt{\alpha 1\cdot\alpha 2\cdot(R0-1) - \left(\alpha 2\cdot\frac{R1}{2}\right)^2}$, and where R1 = $\frac{\alpha 1\cdot R0 + \alpha 2 - 1}{\alpha 1 + \alpha 2 - 1}$.

Now proceed to determine approximations of the covariances of the diffusion approximation.

Covariances of x[i]x[j] are determined by cov1:

```
> cov1:=proc(i,j,tab)
     local x,n;
     x:=op(2,eval(tab));
     add(lhs(x[n])[i]*lhs(x[n])[j]*rhs(x[n]),n=1..nops(x));
  end proc:
```

The local covariance matrix S is determined by the procedure "cov":

```
> cov:=proc(tab)
     local i,j,d,S;
     d:=nops(lhs(op(2,eval(tab))[1]));
     for i from 1 to d do
       for j from 1 to d do
         S[i,j]:=cov1(i,j,tab);
       od;
     od;
     S:=Matrix(2,S);
  end proc:
```

By using the table of transition rates in "trans", we get

```
> Sx:=simplify(cov(trans));
```

$$Sx := [[2\,\mu\,x1 + \mu\,\alpha 1\,R0\,x1\,x2 + \mu\,\alpha 2 - \mu\,\alpha 2\,x1 - \mu\,\alpha 2\,x2 + \mu\,x2, -\mu\,\alpha 1\,R0\,x1\,x2], [-\mu\,\alpha 1\,R0\,x1\,x2, \mu\,\alpha 1\,x2 + \mu\,\alpha 1\,R0\,x1\,x2]] \quad (9)$$

Evaluate the local covariance matrix at the critical point:

```
> S:=simplify(subs(x1=x10,x2=x20,Sx));
```

$$S := \begin{bmatrix} \frac{2\,\mu\,(\alpha 1 + \alpha 2 - 1 + \alpha 1\,\alpha 2\,R0 - \alpha 1\,\alpha 2)}{R0\,(\alpha 1 + \alpha 2 - 1)} & -\frac{\mu\,\alpha 1\,\alpha 2\,(R0-1)}{R0\,(\alpha 1 + \alpha 2 - 1)} \\ -\frac{\mu\,\alpha 1\,\alpha 2\,(R0-1)}{R0\,(\alpha 1 + \alpha 2 - 1)} & \frac{2\,\mu\,\alpha 1\,\alpha 2\,(R0-1)}{R0\,(\alpha 1 + \alpha 2 - 1)} \end{bmatrix} \quad (10)$$

Now proceed to solve A=-S, where A=B*SIG+SIG*BT, and where BT=Transpose(B).

First introduce notation for the elements of the matrix SIG:

```
> SIG:=Matrix(2,[s11,s12,s21,s22]);
```

$$SIG := \begin{bmatrix} s11 & s12 \\ s21 & s22 \end{bmatrix} \quad (11)$$

Next, determine the matrix A:

```
> A:=Matrix(evalm(B&*SIG+SIG&*Transpose(B)));
```

$$A := \left[\left[-\frac{2\,\mu\,\alpha 2\,(\alpha 1\,R0+\alpha 2-1)\,s11}{\alpha 1+\alpha 2-1} + (-\mu\,\alpha 1-\mu\,\alpha 2+\mu)\,s21 + s12\,(-\mu\,\alpha 1-\mu\,\alpha 2+\mu),\ -\frac{\mu\,\alpha 2\,(\alpha 1\,R0+\alpha 2-1)\,s12}{\alpha 1+\alpha 2-1} + (-\mu\,\alpha 1-\mu\,\alpha 2+\mu)\,s22 + \frac{\mu\,\alpha 1\,\alpha 2\,(R0-1)\,s11}{\alpha 1+\alpha 2-1}\right], \left[\frac{\mu\,\alpha 1\,\alpha 2\,(R0-1)\,s11}{\alpha 1+\alpha 2-1} - \frac{s21\,\mu\,\alpha 2\,(\alpha 1\,R0+\alpha 2-1)}{\alpha 1+\alpha 2-1} + (-\mu\,\alpha 1-\mu\,\alpha 2+\mu)\,s22,\ \frac{\mu\,\alpha 1\,\alpha 2\,(R0-1)\,s12}{\alpha 1+\alpha 2-1} + \frac{s21\,\mu\,\alpha 1\,\alpha 2\,(R0-1)}{\alpha 1+\alpha 2-1}\right]\right] \quad \textbf{(12)}$$

Solve the 4 scalar equations gotten from the matrix equation A+S=0 for the 4 unknowns in the matrix Σ:

```
> solve(convert(S+A,set),convert(SIG,set));
```

$$\left\{s11 = \frac{-\alpha 1-\alpha 2+\alpha 2^2+\alpha 1^2+\alpha 1\,\alpha 2+\alpha 1\,\alpha 2\,R0}{\alpha 2\,R0\,(\alpha 1\,R0+\alpha 2-1)},\ s12 = -\frac{1}{R0},\ s21 = -\frac{1}{R0},\ s22 = \left(-R0\,\alpha 1^2+\alpha 2+\alpha 1^2+\alpha 1^2\,R0^2\,\alpha 2+\alpha 1^2\,R0\,\alpha 2-2\,\alpha 2^2+\alpha 2^3-2\,\alpha 1\,\alpha 2\,R0 - \alpha 1^2\,\alpha 2-\alpha 1^3+\alpha 1^3\,R0+2\,\alpha 1\,\alpha 2^2\,R0\right) \Big/ \left(R0\,\left(-4\,\alpha 1\,\alpha 2+2\,\alpha 1^2\,R0\,\alpha 2 + \alpha 1\,\alpha 2^2\,R0-2\,\alpha 1\,\alpha 2\,R0+\alpha 1^2\,\alpha 2+2\,\alpha 1\,\alpha 2^2-1-\alpha 1^2+2\,\alpha 1-3\,\alpha 2^2+3\,\alpha 2 + \alpha 1^3\,R0-2\,R0\,\alpha 1^2+\alpha 1\,R0+\alpha 2^3\right)\right)\right\} \quad \textbf{(13)}$$

```
> assign(%);
```

In simplified expressions below, we use two parameters α and R1 as follows:

```
> alpha=alpha1+alpha2-1;
  R1=(alpha1*R0+alpha2-1)/alpha;
```

$$\alpha = \alpha 1+\alpha 2-1$$

$$R1 = \frac{\alpha 1\,R0+\alpha 2-1}{\alpha} \quad \textbf{(14)}$$

Next, we give alternate and simplified exact expressions for the two variances s11 and s22.
These expressions have been derived "by hand". We use Maple only to check that s11alt=s11 and that s22alt=s22.

```
> s11alt:=1/R0+alpha1/(alpha2*R0*R1);
  simplify(subs(R1=(alpha1*R0+alpha2-1)/(alpha1+alpha2-1),s11-
  s11alt));
```

$$s11alt := \frac{1}{R0}+\frac{\alpha 1}{\alpha 2\,R0\,R1}$$

$$0 \quad \textbf{(15)}$$

```
> s22alt:=alpha1^2*(R0-1)/(alpha^2*R0*R1) + alpha2*R1/(alpha*R0);
  simplify(subs(R1=(alpha1*R0+alpha2-1)/alpha,alpha=alpha1+alpha2
  -1,s22-s22alt));
```

$$s22alt := \frac{\alpha 1^2\,(R0-1)}{\alpha^2\,R0\,R1} + \frac{\alpha 2\,R1}{\alpha\,R0}$$

$$0 \qquad (16)$$

The covariances s12 and s21 are seen above to be simple.
We give them below, and also the one-term asymptotic approximations of the variances s11 and s22 for large alpha1.

```
> s11a:=op(1,asympt(s11,alpha1));
  s12:=s12;
  s21:=s21;
  s22a:=op(1,asympt(s22,alpha1));
```

$$s11a := \frac{\alpha 1}{\alpha 2\,R0^2}$$

$$s12 := -\frac{1}{R0}$$

$$s21 := -\frac{1}{R0}$$

$$s22a := \frac{R0-1}{R0^2} \qquad (17)$$

μI = x20*N divided by σI=$\sqrt{N^* s22}$ is called ρI.
We determine an approximation of ϱI for large α1:

```
> x20;
  x20a:=op(1,asympt(x20,alpha1));
  rhoI:=simplify(x20a*N/sqrt(s22a*N)) assuming R0>1;
```

$$\frac{\alpha 2\,(R0-1)}{R0\,(\alpha 1+\alpha 2-1)}$$

$$x20a := \frac{\alpha 2\,(R0-1)}{R0\,\alpha 1}$$

$$rhoI := \frac{\alpha 2\,\sqrt{R0-1}\,\sqrt{N}}{\alpha 1} \qquad (18)$$

μS = x10*N divided by σS = $\sqrt{N \cdot s11}$ is called ρS.
We determine an approximation of ρS for large α1.

```
> x10;
  simplify(x10*N/sqrt(N*s11a)) assuming R0>1,alpha1>0,alpha2>0;
```

$$\frac{1}{R0}$$

(19)

$$\frac{\sqrt{N}\,\sqrt{\alpha 2}}{\sqrt{\alpha 1}} \tag{19}$$